The biochemical resolving power of fluorescence lifetime imaging

Andrew *L*. Trinh[1,*] and Alessandro Esposito[1]
[1]MRC Cancer Unit, University of Cambridge, Cambridge, United Kingdom
*at862@cam.ac.uk

Keywords: fluorescence lifetime, resolution, microscopy, biochemistry

Published in bioRxiv: doi_________

In this Notebook, we retrace the mathematical formalism presented in Kollner and Wolfrum (https:/-/doi.org/10.1016/0009-2614(92)87068-Z)
We will generate Fig. 1 of their paper to provide an example of how to evaluate Fisher information and to test our analytical techniques used for the case of Gaussian IRFs in a different file.

First, we initialize the Notebook and some parameters as defined in Kollner and Wolfrum.

In[∘]:=
```
(* Define subcripted symbols *)
<< Notation`
Symbolize[ σ_ ]

(* Define positivity of all variables, and set substitution rules *)
opts = {B > 0, α > 0, m > 1, i > 1, τ > 0, A > 0, T > 0, C > 0, t ≥ 0, p > 0};
subs = {τ → 1/α, A → B α};
subsinv = {α → 1/τ, B -> A/α};
```

We then make a convolution integral (f1) of a Dirac excitation pulse with an exponential decay starting from the origin. The number of photons emitted from t=0, to a generic time is the integral Λ[t].

In[∘]:=
```
f1 = Convolve[DiracDelta[x], UnitStep[x] A Exp[-x / τ], x, y]
Λ[t_] = Integrate[f1 //. subs, {y, 0, t}, Assumptions → opts]
```

Out[∘]= $A\, e^{-\frac{y}{\tau}}\, \text{UnitStep}[y]$

Out[∘]= $B - B\, e^{-t\,\alpha}$

The number of photons detected within a time bin is G[i]=Λ[i T]-Λ[(i-1)T], where i is the index bin and T is the bin width. With G[i] we indicate the expectation of Λ[i T]-Λ[(i-1)T], which is otherwise explicitly indicated with the analytical form of the integral Λ[t] to permit derivation of Λ[t] but not of the expectations.

In[∘]:= $\text{LogL}[B\_, \alpha\_] = \sum_{i=1}^{m} \text{Log}\left[\Lambda[i\,T] - \Lambda[(i-1)\,T]\right] G[i] - \sum_{i=1}^{m} \left(\Lambda[i\,T] - \Lambda[(i-1)\,T]\right)$

Out[∘]= $-B + B\, e^{-m\,T\,\alpha} + \sum_{i=1}^{m} G[i]\, \text{Log}\left[B\, e^{-(-1+i)\,T\,\alpha} - B\, e^{-i\,T\,\alpha}\right]$

The Fisher information matrix can be defined like

$$I = \begin{pmatrix} a & c \\ c & b \end{pmatrix}$$

where a, b, and c are:

In[∘]:=
```
a =
 Simplify[-D[LogL[B, α], {α, 2}] //. {G[i] → (Λ[i T] - Λ[(i - 1) T])}, Assumptions → opts]
b = Simplify[-D[LogL[B, α], {B, 2}] //. {G[i] → (Λ[i T] - Λ[(i - 1) T])},
  Assumptions → opts]
c = Simplify[-D[LogL[B, α], {B, 1}, {α, 1}] //. {G[i] → (Λ[i T] - Λ[(i - 1) T])},
  Assumptions → opts]
```

Out[∘]= $-\frac{B\, e^{-m T \alpha} \left(-e^{(1+m) T \alpha} + m^2 + e^{2 T \alpha} m^2 + e^{T \alpha} \left(1 - 2 m^2\right)\right) T^2}{\left(-1 + e^{T \alpha}\right)^2}$

Out[∘]= $\frac{1 - e^{-m T \alpha}}{B}$

Out[∘]= $e^{-m T \alpha}\, m\, T$

Note that we have substituted G[i]→(Λ[i T]-Λ[(i-1) T]) only after differentiation as these now are the expectations. Next, the standard deviation on the parameter α can be evaluated from the top-left element of I^T, and σ_τ can be evaluated by error propagation from σ_α.

In[∘]:=
```
σα = Sqrt[b/(a b - c^2)] //. subsinv
στ = Simplify[Sqrt[((D[τ /. subs[[1]], {α, 1}])^2 /. subsinv)] σα, Assumptions → opts]
```

Out[∘]= $\sqrt{\frac{1 - e^{-\frac{m T}{\tau}}}{A \left(-e^{-\frac{2 m T}{\tau}} m^2 T^2 - \frac{e^{-\frac{m T}{\tau}} \left(1 - e^{-\frac{m T}{\tau}}\right) \left(-e^{\frac{(1+m) T}{\tau}} + m^2 + e^{\frac{2 T}{\tau}} m^2 + e^{T/\tau} \left(1 - 2 m^2\right)\right) T^2}{\left(-1 + e^{T/\tau}\right)^2}\right) \tau}}$

Out[∘]= $\sqrt{\frac{e^{\frac{m T}{\tau}} \left(-1 + e^{T/\tau}\right)^2 \left(-1 + e^{\frac{m T}{\tau}}\right) \tau^3}{A \left(e^{T/\tau} + e^{\frac{T + 2 m T}{\tau}} - e^{\frac{m T}{\tau}} m^2 - e^{\frac{(2+m) T}{\tau}} m^2 + 2 e^{\frac{(1+m) T}{\tau}} \left(-1 + m^2\right)\right) T^2}}$

We can now define the F^2-value. Note that with C we refer to the total number of photon counts actually measured. C is equal to the integral Λ[m T], fact that can be used to replace the amplitude A.

FullSimplify can be used to return a compact equation as per Kollner and Wolfrum. However, we also define F2b. F2b is the very same quantity but its extended analytical form avoids undetermined operations of the type 0/0 or 1/0.

In[∘]:=
```
F2[τ_, m_, T_] = FullSimplify[
  C στ^2/τ^2 //. {Solve[(Λ[m T] //. subsinv) == C, A] [[1]][[1]]}, Assumptions → opts]
F2b[p_, m_] = ExpandAll[
  C στ^2/τ^2 //. {Solve[(Λ[m T] //. subsinv) == C, A] [[1]][[1]]} //. {τ -> m T p^-1}];
(* Expand All is used to avoid numerical issues with Mathematica *)
```

Out[∘]=

$$\frac{\left(-1+e^{T/\tau}\right)^2\left(-1+e^{\frac{m T}{\tau}}\right)^2\tau^2}{\left(e^{T/\tau}+e^{\frac{T+2 m T}{\tau}}-e^{\frac{m T}{\tau}}m^2-e^{\frac{(2+m)T}{\tau}}m^2+2\,e^{\frac{(1+m)T}{\tau}}\left(-1+m^2\right)\right)T^2}$$

In[∘]:=

Next we generate Fig. 1 from Kollner and Wolfrum.

In[∘]:=
```
mlist = 2^Table[m, {m, 1, 10}];
LogLinearPlot[Evaluate@F2b[p, mlist],
 {p, 1., 1000.}, PlotRange → {0, 10}, PlotLabels → mlist]
```

Out[∘]=



In[∘]:=
```
(* This is the photon-efficiency *)
LogLinearPlot[1 / F2[τ, {2, 16, 64, 1024}, .1], {τ, 0.01, 100}, PlotRange → {0, 1}]
```

Out[∘]=



In[∘]:=